

Tipps & Tricks - Nützliche Lösungen zu Problemen und Fragestellungen bei der Umstellung auf die Versionen 8 und 8.1 in Form von Kurzvorträgen

Carina Ortseifen

Universität Heidelberg
Universitätsrechenzentrum
69120 Heidelberg

E-Mail: Carina.Ortseifen@urz.uni-heidelberg.de

Grischa Pfister

Info Ware
69123 Heidelberg

E-Mail: Grischa_Pfister@Info-Ware.de

Heinrich Stürzl

Dade Behring Marburg GmbH
35041 Marburg

E-Mail: Heinrich_Stuerzl@DadeBehring.com

Abstract

Die Neuerungen der SAS Versionen 8 und 8e sind beeindruckend und wer möchte nicht sofort damit arbeiten? Aber was ist mit den alten Dateien, Katalogen und Programmen? Läuft alles noch so wie vorher? Oder muss ich doch Hand anlegen und einiges umstellen?

Die Neuerungen der SAS Versionen 8 und 8e sind beeindruckend und wer möchte nicht sofort damit arbeiten? Aber was ist mit den alten Dateien, Katalogen und Programmen? Läuft alles noch so wie vorher? Oder muss ich doch Hand anlegen und einiges umstellen?

In dieser Sitzung möchten wir einige aus unserer Sicht interessante Neuerungen vorstellen und Hinweise geben, wie Sie die Klippen der neuen Versionen umschiffen, um möglichst schnell von den Neuerungen profitieren zu können.

Ausblick auf die nächste KSFE

Kennen Sie einen interessanten Tipp, den Sie auf der nächsten KSFE gerne vorstellen möchten? So wenden Sie sich bitte an Carina Ortseifen

Themenübersicht

1. Verwirrung in den Dateiendungen –
Was verbirgt sich hinter sd2, sd7 und sas7bdat? Zugriff auf V6-Dateien und -Kataloge
2. Lange Variablennamen -
Probleme bei Makros, die auf automatische SAS-Variablen zugreifen
3. Neue nationale Datumsformate und die Erstellung eigener Datums-, Zeit- und Datum-Zeit-Formate
4. SAS Output als PDF Datei
5. ODS austricksen
6. Export von SAS Grafiken nach Microsoft Office 97 mit Version 8 –
Ein Vergleich von EMF und CGM

1. Verwirrung in den Dateinamenerweiterungen Was verbirgt sich hinter sd2, sd7 und sas7bdat?

Carina Ortseifen

Aufgabenstellung

Mit der Standardeinstellung des Windows-Explorers fällt dem SAS Anwender die Vielfalt vermutlich nicht auf. Doch wer hinter die Kulissen schaut und sich (über die Explorer-Optionen) die *Dateinamenerweiterung* auch *von bekannten Dateitypen anzeigen* lässt, bemerkt bei der Umstellung von SAS Version 6 auf Version 8 eine Änderung in der Dateinamenerweiterung von SAS Tabellen und SAS Katalogen.

Hatten die SAS Tabellen (data sets) in der Version 6 die Erweiterung sd2 und die Kataloge sc2, so heißen diese in der Version 8 jetzt sas7bdat und sas7bcat oder sd7 und sc7. Warum es zwei verschiedene Dateinamenerweiterungen gibt, soll im folgenden erläutert werden.

Zwei verschiedene Dateinamenerweiterungen in der Version 8

Die Verwendung der verschiedenen Erweiterungen hängt vom verwendeten Dateisystem und dessen Einstellung ab. Während NTFS Laufwerke standardmäßig lange Dateinamen und Erweiterungen zulassen, sind die Namen bei FAT und älteren Novell-Laufwerke in der Regel auf acht+drei Zeichen beschränkt. Das SAS System verwendet in diesen beiden Fällen einmal die acht Zeichen langen Erweiterungen sas7bdat und sas7bcat, zum anderen die drei Zeichen langen Erweiterungen sd7 und sc7.

Probleme mit diesen unterschiedlichen Erweiterungen treten dann auf, wenn man in einer SAS Bibliothek sowohl auf SAS Dateien mit kurzer und solche mit langer Erweiterung zugreifen will, was z.B. beim Kopieren der Dateien oder beim Verketteten von SAS Bibliotheken passieren kann.

Option SHORTFILEEXT

Einen Ausweg bietet die Option SHORTFILEEXT der Anweisung LIBNAME. Diese überschreibt die Standardoption LONGFILEEXT und weist das System an, die Dateien mit kurzer Erweiterung zu erkennen.

```
LIBNAME kurz "C:\sasdaten" SHORTFILEEXT;
```

Die Bibliothek "kurz" findet damit alle Dateien mit der Erweiterung sd7 und sc7. Befinden sich in der Bibliothek auch sas7bdat und sas7bcat Dateien, muss eine zweite Libname Anweisung (ohne die Option SHORTFILEEXT) verwendet werden.

Innerhalb der SAS Umgebung können die Dateien mit kurzer Erweiterung mithilfe der Prozedur COPY in Dateien mit langer Erweiterung konvertiert werden (oder umgekehrt). Ausserhalb genügt es, die Dateierweiterung mithilfe des Windows-Explorers abzuändern.

Quelle

- SAS Note SN-002555

2. Zugriff auf V6 Dateien und V6 Kataloge

Carina Ortseifen

Aufgabenstellung

Mit der Installation der Version 8 der SAS Software ist es noch lange nicht getan. Erst wenn auf die "alten" Tabellen (data sets) und Kataloge erfolgreich zugegriffen werden kann und die bestehenden Programme laufen, kann die Version 6 deinstalliert werden. Wie man von der Version 8 auf Version 6 Tabellen und Kataloge zugreifen kann, soll im folgenden beschrieben werden.

SAS Tabellen

Prinzipiell kann mit der Version 8 auf alle SAS Tabellen zugegriffen werden, die in der Version 6 angelegt worden sind. Eventuell muss jedoch durch eine geeignete Option in der Anweisung LIBNAME dafür Sorge getragen werden, dass das SAS System diese Tabellen erkennt.

Hierbei ist zu unterscheiden, ob die Tabellen der Versionen 6 und 8 in verschiedenen Ordnern oder in einem gemeinsamen Ordner liegen. Im ersten Fall, wenn die Tabellen in verschiedenen Ordnern liegen, genügt es, jeweils eine LIBNAME Anweisung für den Ordner mit den Version 6 Tabellen auszuführen und eine LIBNAME Anweisung für die Version 8 Tabellen. Im SAS Explorer werden alle Tabellen angezeigt und können wie gewohnt verarbeitet werden.

Liegen die Tabellen jedoch in einem gemeinsamen Ordner, müssen ebenfalls zwei LIBNAME Anweisungen gesetzt und zusätzlich einmal die ENGINE Option verwendet werden. Das folgende Beispiel möge dies verdeutlichen. Im Ordner "daten" auf Laufwerk C: befinden sich sowohl V6 als auch V8 Dateien.

```
LIBNAME daten6 v6 "C:\daten";  
LIBNAME daten8 "C:\daten";
```

Mit der Bibliothek daten6 werden die SAS Dateien der Version 6 angezeigt, mit der Bibliothek daten8 die Dateien der Version 8.

Falls V6 Tabellen in V8 Tabellen konvertiert werden sollen, so kann dies auch einfach per Drag&Drop geschehen. Dazu im SAS Explorer die gewünschten Tabellen in der V6 Bibliothek markieren und mit der linken Maustaste in eine V8 Bibliothek ziehen.

Dies funktioniert umgekehrt prinzipiell genauso, wobei dadurch V8 Funktionalität wie Variablenamen mit mehr als 8 Zeichen, Charaktervariablen mit einer Länge > 200 Zeichen, etc. verloren gehen!

SAS Kataloge

In SAS Katalogen werden Formate, Grafiken, AF-Anwendungen u.v.a.m. gespeichert. Im Unterschied zu den Tabellen können V6 Kataloge nicht ohne Konvertierung in der Version 8 korrekt gelesen werden. Die Konvertierung kann mit der Prozedur DATASETS und der Anweisung COPY (oder der Prozedur COPY) erfolgen oder mit den Prozeduren CPORT und CIMPORT. In allen Fällen muss zunächst in der Version 6 des SAS Systems ein Transportfile erstellt werden, der dann in der Version 8 importiert wird.

Eine Konvertierung per Drag&Drop ist zwar möglich, kann jedoch zu fehlerhaften Katalogen führen und ist deshalb nicht empfehlenswert.

Beispiel für den Gebrauch der Prozeduren CPORT und CIMPORT:

Im Ordner C:\daten liegt der SAS Katalog projekt.sc2. Dieser wird zunächst in der Version 6 in einen Transportfile, C:\trans.fer, konvertiert:

```
LIBNAME daten "C:\daten";  
FILENAME trans "C:\trans.fer";  
PROC CPORT CAT=daten.projekt FILE=trans;  
RUN;
```

Dieser Transportfile wird mit folgenden Anweisungen in der Version 8 eingelesen:

```
LIBNAME daten "C:\daten8";  
FILENAME trans "C:\trans.fer";  
PROC CIMPORT CAT=daten.projekt INFILE=trans;  
RUN;
```

Für Format-Kataloge gibt es zusätzlich die Möglichkeit, mithilfe der Optionen CNTLIN und CNTLOUT der Prozedur FORMAT einzelne Formate zu konvertieren.

3. Lange Variablenamen - Probleme bei Makros, die auf automatische SAS-Variablen zugreifen

Heinrich Stürzl

Einleitung

Eines der wesentlichen Kennzeichen der Version 8 ist die Möglichkeit, Variablenamen mit mehr als acht Zeichen zu verwenden. Die Namen von Variablen dürfen nun bis zu **32 Zeichen** lang sein, wobei die übrigen Regeln bei der Namensbildung (nur Ziffern, Buchstaben, Unterstrich und keine Ziffer am Anfang) weiterhin gelten. Damit verbunden ist eine zweite einschneidende Änderung.

Bei Variablen wird die **Groß/Kleinschreibung des Namens (Mixed Case)** gespeichert und von den Prozeduren verwendet. Maßgeblich ist die Schreibweise bei der Definition einer neuen Variable, also z. B. bei einer Zuweisung, oder den Anweisungen INPUT, LENGTH, ARRAY etc. Trotzdem kann auf die Variablen in jeder beliebigen Schreibweise zugegriffen werden. Zum Beispiel wird auf die Variable „PatID“ mit „patid“, „Patid“ oder „PATID“ etc. zugegriffen. Die genaue Schreibweise des Namens wird in der SAS Tabelle (SAS data set) gespeichert und von allen Prozeduren genauso wiedergegeben.

Die Möglichkeit lange Variablenamen und damit auch die Mixed Case Funktionalität zu verwenden, wird über die neue Systemoption **VALIDVARNAME** gesteuert. Standardmäßig eingestellt und empfohlen wird **VALIDVARNAME=V7**, wodurch die neuen Möglichkeiten aktiviert werden. Mit **VALIDVARNAME=V6** bleiben die alten Regeln gültig, wobei die Namen aller Variablen in Großbuchstaben gespeichert und ausgegeben werden. **VALIDVARNAME=UPCASE** erlaubt zwar lange Namen, die jedoch in Großbuchstaben gespeichert werden.

Die Standardeinstellung **VALIDVARNAME=V7** bewirkt, dass der Prozeduroutput anders aussieht als gewohnt (auch dann, wenn man keine langen Variablenamen benutzt!). Die Prozedur PRINT beispielsweise bricht Variablenamen um, wenn der Name länger als die benötigte Spaltenbreite ist und im Namen auf einen Kleinbuchstaben ein Großbuchstabe folgt.

Diese Einstellung kann aber auch gravierende Konsequenzen haben, die zum Beispiel dazu führen können, dass bewährte Makros in der Version 8 nicht mehr funktionieren! Dies soll am Beispiel von PROC REG demonstriert werden, wobei die Namen aller benutzerdefinierten Variablen kürzer als acht Zeichen sind.

Beispiel

```

OPTIONS VALIDVARNAME=V7;

DATA eins;
INPUT
  Tag      Geraet  Wert;      DATALINES;
  1        1      100
  1        2      200
  1        3      300
  2        1      400
  2        2      500
  2        3      600
  3        1      700
  3        2      800
  3        3      900 ;
RUN;

PROC REG DATA=eins OUTEST=zwei;
  MODEL geraet=wert;
  BY tag;
RUN;

PROC PRINT DATA=zwei;
RUN;
    
```

Ergebnis mit Option VALIDVARNAME=V6 (entspricht Version 6 Output)

OBS	TAG	_MODEL_	_TYPE_	_DEPVAR_	_RMSE_	INTERCEP	WERT	GERAET
1	1	MODEL1	PARMS	GERAET	0	0	0.01	-1
2	2	MODEL1	PARMS	GERAET	0	-3	0.01	-1
3	3	MODEL1	PARMS	GERAET	0	-6	0.01	-1

Ergebnis mit Option VALIDVARNAME=V7

Obs	Tag	_MODEL_	_TYPE_	_DEPVAR_	_RMSE_	Intercept	Wert	Geraet
1	1	MODEL1	PARMS	Geraet	0	-0	0.01	-1
2	2	MODEL1	PARMS	Geraet	0	-3	0.01	-1
3	3	MODEL1	PARMS	Geraet	0	-6	0.01	-1

Ergebnis mit Option VALIDVARNAME=UPCASE

OBS	TAG	_MODEL_	_TYPE_	_DEPVAR_	_RMSE_	INTERCEPT	WERT	GERAET
1	1	MODEL1	PARMS	GERAET	0	0	0.01	-1
2	2	MODEL1	PARMS	GERAET	0	-3	0.01	-1
3	3	MODEL1	PARMS	GERAET	0	-6	0.01	-1

Zusammenfassung

Die Einstellung der Option VALIDVARNAME kann entscheidenden Einfluss auf die Erstellung von automatisch erstellten Output Data Sets haben. Dies kann sich sowohl auf die Namen als auch auf die Ausprägungen von automatischen Variablen auswirken.

Beispiel PROC REG:

1. Bei der Berechnung einer linearen Regressionsgeraden speichert PROC REG den Ordinatenabschnitt in einer Variablen „Intercept“ bzw. „INTERCEPT“ statt wie bisher „INTERCEP“.
2. Der Name der abhängigen Variable wird von PROC REG wie bisher in der alphanumerischen Variable _DEPVAR_ gespeichert, wobei die Ausprägung sich danach richtet mit welcher Schreibweise die abgebildete Variable gespeichert ist. Die Länge von _DEPVAR_ richtet sich automatisch nach der Länge des Namens der abgebildeten Variable.

4. Neue nationale Datumsformate und die Erstellung eigener Datums-, Zeit- und Datum-Zeit-Formate

Carina Ortseifen, Grischa Pfister

Aufgabenstellung

Mit der Version 8 der SAS Software wurden zahlreiche neue Datumsformate eingeführt, auf die viele SAS Anwender schon lange gewartet haben. Außerdem kann der Anwender sich eigene Formate "basteln".

Neue Datumsformate

Während es beim Einlesen von Datumsangaben, also beim Informat, keine Rolle spielte, welches Trennzeichen zwischen Tag, Monat und Jahr stand, wurden die Zahlenwerte in der Ausgabe mit dem Format DDMMYYw. in der Version 6 des SAS Systems stets durch einen Schrägstrich getrennt, wie dies im US-amerikanischen Raum üblich ist.

Bei den neuen Datumsformaten der Version 8 gibt es nun die Möglichkeit, das Trennzeichen zwischen Tag, Monat und Jahr festzulegen. Folgende Trennzeichen sind möglich: Leerzeichen, Doppelpunkt, Querstrich, Punkt, Schrägstrich oder kein Trennzeichen. Die zugehörigen Formate erhalten damit folgende Namen: (In der rechten Spalte wird das damit erzielte Format dargestellt.)

DDMMYY B w.	Leerzeichen (Blank)	08 03 2001
DDMMYY C w.	Doppelpunkt (Colon)	08:03:2001
DDMMYY D w.	Querstrich (Dash)	08-03-2001
DDMMYY N w.	Kein Trennzeichen (None)	08032001
DDMMYY P w.	Punkt (Period)	08.03.2001
DDMMYY S w.	Schrägstrich (Slash)	08/03/2001

Eigene Datums-, Zeit- und Datum-Zeit-Formate erstellen

Ebenfalls neu in Version 8 ist die Möglichkeit, eigene Datumsformate zu erstellen. Damit haben wir Benutzer die Möglichkeit, die Darstellung komplett nach unseren eigenen Vorstellungen zu gestalten.

Dazu wird die PICTURE-Anweisung der Prozedur FORMAT benötigt. Mit dieser Anweisung können Schablonen für die Zahlendarstellung erzeugt werden, sie ist in Version 8 um einige interessante Punkte erweitert worden. Für die Darstellung von Datums-, Zeit- oder Datum-

Zeit-Werten bedeutet dies: Es wurde eine Reihe von Platzhaltern eingeführt, die jeweils für eine bestimmte Darstellungsform eines Wertes stehen. So kann ein Wochentag z.B. durch eine Zahl oder den ausgeschriebenen Namen des Tages dargestellt werden etc.

Beispiel 1:

```
PROC FORMAT;
  PICTURE datum
  other = '%A, %B, %Y' (datatype=date);
run;

DATA _null_;
  datum = today();
  put datum = : datum30.;
run;
```

Die Picture-Option „(datatype=date)“ sorgt dafür, dass das Format als Datumsformat erkannt wird. „%A“ ist der Platzhalter für den ausgeschriebenen Tag, „&B“ steht für den ausgeschriebenen Monat und „%Y“ für das vierstellige numerische Jahr.

Das Ergebnis ist:

```
datum=Fri day, March, 2001
```

Formate, die auf diese Weise definiert werden, unterstützen automatisch den „national language support“ von SAS, d.h. über die Option „dflang = sprache“ kann die Sprache für die Darstellung geändert werden. Obiges Beispiel mit der Einstellung „dflang = ger“ führt dann zu folgender Darstellung:

```
datum=Frei tag, März, 2001
```

Die Platzhalter können beliebig mit Standardtext kombiniert werden, wichtig ist nur, dass das Format genug Platz bietet, um den gesamten Text darstellen zu können.

Beispiel 2:

```
PROC FORMAT;
  PICTURE datum (default = 40)
  other = 'Heidelberg, %A den %0d.%0m.%0y' (datatype = date);
run;
Options dflang = ger;
DATA _null_;
  datum = today();
  put datum = : datum.;
run;
```

„%d“, „%m“ und „%y“ sind die Platzhalter für die zweistellige numerische Darstellung von Tag, Monat und Jahr, die eingefügte Null sorgt dafür, dass einstellige Zahlen mit führender Null dargestellt werden.

Das Ergebnis ist:

```
datum=Hei del berg, Frei tag den 16.03.01
```

Es gibt eine Vielzahl von vordefinierten Platzhaltern, eine komplette Liste findet sich in der Online-Hilfe unter der Prozedur FORMAT (z.B. mit dem Kommando `help format`) bei der PICTURE-Anweisung.

Auch die Option `dflang` ist in der Online-Hilfe vertreten, sie steht unter den System-Optionen, auch hier kommt man mit dem Kommando „`help dflang`“ am schnellsten zur gewünschten Information.

5. SAS Output als PDF Datei

Heinrich Stürzl

Aufgabenstellung

PDF = Portable Document Format von Adobe ist das gängige Format zur Weitergabe von Dokumenten per Email und im Internet. Dies ist bedingt durch seine Vorteile:

- PDF Dokumente können auf beliebigen Plattformen gelesen und gedruckt werden unabhängig von der Hardware, auf der sie erstellt worden sind.
- Zum Lesen genügt der Adobe Acrobat Reader, welcher kostenfrei verfügbar ist.
- Verschiedene PDF Dateien lassen sich zu einer Datei zusammenfügen.
- Es lassen sich Anmerkungen, Inhaltsverzeichnisse hinzufügen.
- PDF Dateien sind relativ klein.
- Empfänger kann PDF Dateien (fast) nicht ändern (Radiergummifunktion).

Erstellung von PDF Dokumenten mit SAS

Die Erstellung von PDF Dateien kann mit Hilfe der Software Acrobat der Fa. Adobe erfolgen. Darin sind zwei verschiedene Möglichkeiten zur Erstellung enthalten.

1. PDFWriter, ein Druckertreiber, der ein Dokument direkt aus der Anwendung konvertiert.
2. Acrobat Distiller konvertiert Postscript (PS) Dateien und erzielt meist bessere Ergebnisse.

Alternativ kann man auch die Software Ghostscript verwenden, welche kostenfrei erhältlich ist unter <http://www.cs.wisc.edu/~ghost/>.

Ab der SAS Version 8.1 ist es möglich, PDF Dateien über ODS PRINTER und die PDF Option auch ohne die Adobe Acrobat Software zu erstellen (noch experimentell). Mit Version 8.2 wird diese Funktionalität produktiv.

In SAS kann grundsätzlich unterschieden werden zwischen den Möglichkeiten mit und ohne ODS (Output Delivery System). Letztere können auch unabhängig von der Version 8 verwendet werden.

Bei PDF Dokumenten gibt es die Möglichkeit, die verwendeten Schriftarten in das Dokument einzubetten. Dies ist vor allem dann sinnvoll, wenn die Schriftart beim Empfänger möglicherweise nicht vorhanden ist, wie dies bei SAS Monospace der Fall sein kann. In diesem Fall werden die Rahmenzeichen beim Empfänger nicht korrekt dargestellt, weil diese Schriftart durch eine Standardschriftart ersetzt wird. Falls man SAS Output mit "SAS Monospace" erstellt, ist es deshalb empfehlenswert diese Schriftart in das PDF Dokument einzubetten. Dies hat allerdings den Nachteil, dass die PDF Datei dadurch größer wird. Die entsprechenden Einstellungen sind sowohl beim PDFWriter (Eigenschaften), als auch im Distiller möglich (Voreinstellungen ⇔ Einstellungen).

Da sich SAS bei der Erzeugung von Listenoutput im Seitenlayout an der verfügbaren LINESIZE und PAGESIZE orientiert, ist es wichtig, dass vor dem Start eines SAS

Programms der gewünschte Drucker bzw. Druckertreiber (PDFWriter oder Distiller) eingestellt wird.

Nachfolgend werden sieben verschiedene Möglichkeiten beschrieben.

Testbedingungen

Zum Test wurde ein einfaches SAS Programm verwendet (siehe letzte Seite dieses Tipps), welches ein einseitiges Datenlisting mit PROC PRINT im Hochformat und zwei Punktwolkengrafiken mit PROC GPLOT per BY Anweisung im Querformat erzeugt.

Um die korrekte Darstellung der ersten und letzten Zeile, sowie der ersten und letzten Spalte im Listing zu überprüfen werden Titel und Fußzeilen linksbündig ausgerichtet und die DATE Option verwendet. Die Titelzeilen enthalten deutsche Umlaute.

Alle Tests wurden mit SAS Version 6.12 (ohne ODS) und 8.1 unter Windows NT 4.0 mit Adobe Acrobat 4.0 (deutsch) durchgeführt, wobei als Schriftart "SAS Monospace" 8 pt, Hochformat und 5 mm Seitenränder oben, unten und rechts, sowie ein Lochungsrand links von 10 mm eingestellt wurden.

In der Grafik werden die SAS Softwarefonts Swiss und Simplex verwendet.

Zum Drucken wurde ein HP Laserjet 5 verwendet.

5.1 Ohne ODS, PDFWriter

1. Im Print Setup Dialog **Acrobat PDFWriter** als Drucker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.

```
OPTIONS SYSPRINT='Acrobat PDFWriter';  
OPTIONS SYSPRINTFONT='SAS Monospace' 8;  
OPTIONS ORIENTATION=PORTRAIT PAPERSIZE=A4 LEFTMARGIN="1.0 CM"  
RIGHTMARGIN="0.5 CM" TOPMARGIN="0.5 CM" BOTTOMMARGIN="0.5 CM";
```


2. Anschließend das Programm starten und den Output im Listen- bzw. Graphfenster erzeugen.
3. Aus dem Listen- bzw. Graphfenster den gewünschten Output "drucken" und dabei einen Namen für die PDF Datei angeben. Für jede Grafik muss eine separate PDF Datei angegeben werden. Das Listenfenster kann entweder als Ganzes oder die Einträge können einzeln in eine PDF-Datei geschrieben werden.

Ergebnis:

- Die Grafiken werden gedreht angezeigt (im Hochformat), jedoch korrekt gedruckt. Nachteil, die Grafiken sind **grob aufgelöst**, Dateigröße je **17 k**
- Liste wird vollständig und mit Umlauten angezeigt und gedruckt, Dateigröße **16 k**, **LS=114, PS=71**.

5.2. Ohne ODS, Distiller

1. Der Distiller darf nicht für einen bestimmten Port (z. B. lpt1), sondern muss für den Anschluss "File" konfiguriert sein. Dazu unter Einstellungen ⇒ Drucker die Distiller Eigenschaften aufrufen und in der Lasche "Anschlüsse" den Eintrag "File" auswählen. Andernfalls erscheint beim Erstellen der Postscript-Datei in SAS die folgende Fehlermeldung (auch mit 8.1).



2. Im Print Setup Dialog **Acrobat Distiller** als Drucker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.

```
OPTIONS SYSPRINT='Acrobat Distiller';
OPTIONS SYSPRINTFONT='SAS Monospace' 8;
OPTIONS ORIENTATION=PORTRAIT PAPERSIZE=A4 LEFTMARGIN="1.0 CM"
RIGHTMARGIN="0.5 CM" TOPMARGIN="0.5 CM" BOTTOMMARGIN="0.5 CM";
```

3. Anschließend das Programm starten und den Output im Listen- bzw. Graphfenster erzeugen.
4. Aus dem Listen- bzw. Graphfenster den gewünschten Output "drucken" und dabei einen Namen für die **Postscript Datei** (*.ps) angeben. Für jede Grafik muss eine separate PS Datei angegeben werden. Das Listenfenster kann entweder als Ganzes oder die Einträge können einzeln in eine PS Datei geschrieben werden.
5. Mit Hilfe des Acrobat Distillers die Postscript Dateien in PDF Dateien umwandeln. Dies kann auch automatisch geschehen, indem im Distiller ein "überwachter Ordner" mit den Unterverzeichnissen "IN" und "OUT" definiert wird. Der Distiller prüft im eingestellten Zeitintervall, ob sich PS Dateien im Eingangsordner befinden und konvertiert diese und stellt sie automatisch in den Ausgangsordner.


Ergebnis:

- Die Grafiken werden gedreht angezeigt (im Hochformat), jedoch korrekt gedruckt. Die Grafiken haben eine deutlich **bessere Auflösung**, welche annähernd so hoch ist wie bei der Original SAS Grafik, Dateigröße je **29 k**
- Liste wird vollständig und mit Umlauten angezeigt und gedruckt, Dateigröße **16 k, LS=115, PS=72**.

ODS PRINTER

Mit Hilfe der ODS PRINTER Anweisung lassen sich hochauflösende Outputs (auch Listenoutput!) erstellen. Listenoutput von PROC PRINT, CONTENTS etc. wird als formatierte Tabelle dargestellt. Diese sehen zwar gut aus, haben aber den Nachteil, dass deutlich weniger Zeilen und Spalten auf eine Seite passen als bei herkömmlichem zeichenorientiertem Listenoutput (ODS LISTING). Das Aussehen der Tabellen kann über die STYLE= Option beeinflusst werden. Standardmäßig wird STYLE=PRINTER verwendet.

Die Liste der verfügbaren Styles kann man anzeigen, indem man im Results Fenster den obersten Eintrag mit der rechten Maustaste anklickt und Templates auswählt.



Eigene Styles lassen sich mit PROC TEMPLATE erstellen.

Über die Option FILE= lässt sich dabei eine Datei angeben, in die der gesamte Output aller Prozeduren in der Reihenfolge ihrer Erzeugung geschrieben wird. Diese umfasst auch By-Group-Processing und Run-Group-Processing. Das Schreiben in die Ausgabedatei bleibt solange wirksam, bis die Printer Destination geschlossen wird (ODS PRINTER CLOSE). In dieser Datei können sowohl Listings als auch Grafiken enthalten sein.

Siehe auch SAS OnlineDoc: Base SAS Software → Guide to the SAS Output Delivery System

5.3 Mit ODS PRINTER, PDFWriter

1. Im Print Setup Dialog **Acrobat PDFWriter** als Drucker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.
2. PDF Datei in ODS PRINTER Anweisung festlegen.

```
ODS PRINTER PRINTER='Acrobat PDFWriter' FILE='C:\temp\test1.pdf';
```

3. Anschließend das Programm starten. Der Listen- und Grafikoutput aller Prozeduren wird in die angegebene PDF Datei geschrieben.

Ergebnis:

- Alle Outputs befinden sich in einer PDF Datei, **86 k** (> 17+16+16=49 k)
- Die Grafiken werden gedreht angezeigt (im Hochformat), jedoch korrekt gedruckt.
- Schriftart Simplex kann nicht verwendet werden
(WARNING: Font specification "simplex" failed. Using default font family instead.)
- Die Grafiken sind ebenfalls **grob aufgelöst**
- Liste wird hochaufgelöst als Tabelle erstellt.
- Umlaute korrekt
- **Am linken und rechten Rand der Liste wird Text abgeschnitten, wenn im Print Setup linke und rechte Ränder eingestellt werden.**
- **Fußzeile fehlt vollständig**, auch wenn keine Ränder eingestellt werden.

5.4 Mit ODS PRINTER, Distiller

1. Im Print Setup Dialog **Acrobat Distiller** als Drucker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.
2. PS Datei in ODS PRINTER Anweisung festlegen.
`ODS PRINTER PRINTER='Acrobat Distiller' FILE='C:\temp\test1.ps';`
3. Anschließend das Programm starten. Der Listen- und Grafikoutput aller Prozeduren wird in die angegebene PS Datei geschrieben.
4. Mit Hilfe des Acrobat Distillers die Postscript Datei in PDF Datei umwandeln.

Ergebnis:

- Alle Outputs befinden sich in einer PDF Datei, **256 (!) k** (> 29+29+16=74 k)
- Die Grafiken werden gedreht angezeigt (im Hochformat), jedoch korrekt gedruckt.
- Schriftart Simplex kann nicht verwendet werden
(WARNING: Font specification "simplex" failed. Using default font family instead.)
- Die Grafiken sind **gut aufgelöst**
- Umlaute korrekt
- Liste wird hochaufgelöst als Tabelle erstellt und kann vollständig angezeigt und gedruckt werden.
- **Nachteil: große Datei.**

5.5 Mit ODS PRINTER PS Option

1. Im Print Setup Dialog den gewünschten Ausgabedruker (muss nicht postscriptfähig sein) auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.
2. PS Datei in ODS PRINTER Anweisung festlegen. Option POSTSCRIPT oder PS.
`ODS PRINTER PS FILE='C:\temp\test2.ps';`
3. Anschließend das Programm starten. Der Listen- und Grafikoutput aller Prozeduren wird in die angegebene PS Datei geschrieben.
4. Mit Hilfe des Acrobat Distillers die Postscript Datei in PDF Datei umwandeln

Ergebnis:

- Alle Outputs befinden sich in einer PDF Datei, **113 k**
- Die Grafiken werden gedreht angezeigt (im Hochformat), jedoch korrekt gedruckt.
- Schriftart Simplex kann nicht verwendet werden
(WARNING: Font specification "simplex" failed. Using default font family instead.)
- Die Grafiken sind **gut aufgelöst**
- **Keine Umlaute in der Liste**, nur in der Grafik korrekt
- Liste wird hochaufgelöst als Tabelle erstellt und kann vollständig angezeigt und gedruckt werden.
- **Nachteil: Der bedruckbare Bereich bei Grafik und Liste ist deutlich verkleinert.**

5.6 Mit ODS PRINTER PDF Option (β in 8.1)

Im Gegensatz zu allen anderen Möglichkeiten der Erzeugung von PDF Dateien wird hierfür die Adobe Acrobat Software nicht benötigt. Diese Option ist in SAS Version 8.1 noch experimentell und fehlerbehaftet; Sie wird produktiv mit Version 8.2

1. Im Print Setup Dialog den gewünschten Ausgabedruker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.
2. PDF Datei in ODS PRINTER Anweisung festlegen. Option PDF
`ODS PRINTER PDF FILE='C:\temp\test2.pdf';`
3. Anschließend das Programm starten. Der Listen- und Grafikoutput aller Prozeduren wird in die angegebene PDF Datei geschrieben.

Ergebnis:

- Alle Outputs befinden sich in einer PDF Datei, **684 (!) k**
- **Die Grafiken werden nicht angezeigt.** Dies soll in 8.2 behoben sein. Fehlermeldung:



- Schriftart Simplex kann nicht verwendet werden
(WARNING: Font specification "simplex" failed. Using default font family instead.)
- **Umlaute in der Liste fehlerhaft.** Dies soll in 8.2 behoben sein (SN-003627).
- Liste wird hochaufgelöst als Tabelle (inkl. Thumbnails) erstellt und kann vollständig angezeigt und gedruckt werden.
- **Nachteil: Datei sehr groß.**
- **Nachteil: Der bedruckbare Bereich bei Grafik und Liste ist deutlich verkleinert.**

5.7 Mit ODS RTF ⇔ Word ⇔ PDF

Eine PDF Datei lässt sich auch auf dem Umweg über eine RTF Datei und Word erstellen. Dazu wird die RTF Datei in Word geöffnet und dort mit Hilfe von Adobe Acrobat direkt in eine PDF umgewandelt.

Die ODS RTF Destination ist mit der SAS Version 8.1 produktiv.

1. Im Print Setup Dialog den gewünschten Ausgabedruker auswählen und gewünschte Schriftart, Schriftgröße und Ränder festlegen. Alternativ kann dies auch per Programm mit der SYSPRINT Option geschehen.

2. RTF Datei in ODS RTF Anweisung festlegen

```
ODS RTF FILE='C:\temp\test2.rtf';
```

3. Anschließend das Programm starten. Der Listen- und Grafikoutput aller Prozeduren wird in die angegebene RTF Datei geschrieben.
4. Die RTF Datei in Word öffnen und eventuell nachbearbeiten.
5. Mit Hilfe von Adobe Acrobat aus Word heraus eine PDF Datei erzeugen über Datei ⇒ Adobe PDF erstellen..., wobei eingestellt werden kann, ob der PDFWriter oder Distiller zur Umwandlung verwendet werden soll.

Ergebnis (Word 97, Distiller mit der Einstellung "MonitorOptimiert"):

- Alle Outputs befinden sich in einer RTF Datei (36 k) bzw. PDF Datei, 17 k
- Die **Grafiken** erscheinen im Querformat, jedoch relativ klein und **mit sehr grober Auflösung**. Jede Grafik erscheint auf einer separaten Seite in einem neuen Abschnitt (auch bei By-Group-Processing).
- Schriftart Simplex, Swiss kann nicht verwendet werden (ersetzt durch SAS Monospace).
WARNING: Font specification "SWISS" failed. Using default font family instead.
WARNING: Font specification "simplex" failed. Using default font family instead.
- Liste wird hochaufgelöst als Tabelle erstellt und kann vollständig angezeigt und gedruckt werden.

Zusammenfassung

Tests mit SAS Version 8.1

Ohne ODS Einzeldateien	Grafik	Listing (zeichenorientiert)	Dateigröße
1. PDFWriter	Geringe Auflösung	✓	49 k (Summe)
2. Distiller	✓	✓	74 k (Summe)

ODS PRINTER Eine Gesamtdatei	Grafik	Listing (hochauflösend)	Dateigröße
3. PDFWriter	Geringe Auflösung	Fußzeile fehlt keine Ränder erlaubt	86 k
4. Distiller	✓	✓	256 k
5. PS Option	Verkleinert	Keine Umlaute Verkleinert	113 k
6. PDF Option (β 8.1)	–	Keine Umlaute Verkleinert	684 k

ODS RTF Eine Gesamtdatei	Grafik	Listing (hochauflösend)	Dateigröße
7. Distiller	Sehr geringe Auflösung	✓	36 k (RTF) 17 k (PDF)

Hinweis: Bezüglich der Größe von PDF Dateien, die mit SAS erstellt worden sind, gibt es die Note SN-003733. Darin wird als Grund für die um den Faktor 3-4 vergrößerten Dateien genannt, dass SAS nicht-optimierten PDF Code erzeugt. Durch ein nachträgliches erneutes Speichern der PDF Dateien mit Adobe Acrobat unter Anwendung der Optimierung können diese verkleinert werden.

Quellen

- SAS Notes (SN-003627, SN-003733)
- FAQ zu ODS PRINTER von SAS Entwickler Brian Schellenberger
<http://www.sas.com/rnd/base/topics/odsprinter/faq.html>
- SEUGI 2000 Vortrag von Lex Jansen „Creating PDF Documents including Hyperlinks, bookmarks and a table of contents with the SAS Software“
<http://www.lex-jansen.demon.nl/seugi>

Testprogramm

```
%LET methode=Acrobat Distiller;

GOPTIONS RESET=ALL KEYMAP=WINANSI DEVMAP=WINANSI DEV=WIN
  TARGET=WINPRTC ROTATE=LANDSCAPE GUNIT=PCT CTEXT=BLACK FTEXT=SWISS
  HTEXT=2.0 CBY=BLACK FBY=SWISS HBY=2.5 LFACTOR=2;

* Standardeinstellung für Seitenlayout festlegen --> LS, PS;
OPTIONS NUMBER DATE NOCENTER SYSPRINT="&methode";
OPTIONS SYSPRINTFONT='SAS Monospace' 8;
OPTIONS ORIENTATION=PORTRAIT PAPERSIZE=A4 LEFTMARGIN="1.0 cm"
  RIGHTMARGIN="0.5 cm" TOPMARGIN="0.5 cm" BOTTOMMARGIN="0.5 cm";

ODS LISTING;

TITLE1 H=4.0 "PDF Output erstellen mit &methode";
TITLE2 H=3.0 "Umlaute öäü ÖÄÜ ß";
FOOTNOTE1 F=SIMPLEX A=90 H=1.5 "HS (&pname) &SYSVLONG &SYSDATE
  %SYSFUNC(TIME(),TIME5.)";

DATA class;
  SET SASHELP.CLASS;
  Groesse=height*2.54; * inch --> cm;
  Gewicht=weight*0.45; * lb --> kg;
  FORMAT groesse 3. gewicht 5.1;
  LABEL Groesse="Körpergröße [cm]" Gewicht="Körpergewicht [kg]";
RUN;

PROC SORT DATA=class; BY sex; RUN;

*X; AXIS1
  LABEL =(H=2.5) OFFSET=(0.4 CM) VALUE =(H=2.0) MAJOR =(H=0.3 CM)
  MINOR =(N=4 H=0.1 CM);

*Y; AXIS2
  LABEL =(H=2.5 A=90) OFFSET=(0.2 CM) VALUE =(H=2.0) MAJOR =(H=0.3
  CM) MINOR =(N=4 H=0.1 CM);

SYMBOL1;
SYMBOL1 C=RED F= V=CIRCLE I=NONE H=2 L=2 WIDTH=2;
SYMBOL2 C=BLACK F= V=DOT I=NONE H=2 L=2 WIDTH=2;
```



```

LEGEND1;
LEGEND1 ACROSS=2 LABEL=NONE OFFSET=(0.9 CM,0) SHAPE=SYMBOL( 18, 2.0)
        POSITION=(BOTTOM CENTER OUTSIDE) MODE=RESERVE FRAME
        VALUE=(H=2.0 F=SWISS J=L);

PROC GPLOT DATA=class;
    PLOT gewicht*grosesse / HAXIS=AXIS1 VAXIS=AXIS2 FRAME
    LEGEND=LEGEND1;
    BY sex;
RUN; QUIT;

PROC PRINT DATA=class LABEL;
    BY sex;
    ID sex;
RUN;

```

6. ODS austricksen

Grischa Pfister

Einleitung

Dieser Tipp bezieht sich auf die Erstellung von vielen HTML-Dateien mit dem Output Delivery System (ODS), eine Aufgabe die meistens im Batch-Modus abgearbeitet wird, und über die z.B. statische Berichtswesen für ein Intranet realisiert werden. Zwar können mit Hilfe des ODS schnell und einfach HTML-Dateien generiert werden, im Detail zeigen sich aber noch gewisse Schwächen.

Das Output Delivery System

Das ODS stellt eine völlig neue Art der Organisation von SAS-Output zur Verfügung, hier wird der Prozedur-Output für alle Prozeduren zentral verwaltet und gelayoutet. Einer der wichtigsten Kanäle ist sicherlich HTML (Hypertext Markup Language), die Aufbereitung von Ergebnissen für das World Wide Web (WWW), daneben gibt es „destinations“ genannte Ausgabekanäle wie PDF, RTF, XML oder PRINTER.

Funktionsweise des ODS

Das ODS schaltet sich zwischen die Erzeugung von Prozedur-Output und die tatsächliche Darstellung. Dabei wirkt es sich auf zwei Arten auf die Darstellung der Ergebnisse aus:

1. Erzeugt eine Prozedur Output, wird dieser zunächst in eine Daten-Komponente gespeichert und diese an das ODS weitergeleitet. Das ODS enthält zu jeder Daten-Komponente eine passende Template-Komponente, die das generelle Output-Format (welche und wie viele Spalten, Beschriftung, Zahlenformate etc.) regelt.
2. Dann wird das Ergebnis mit Hilfe einer Style-Komponente für das jeweilige Ausgabe-Ziel (HTML, RTF, PDF, etc.) gelayoutet.

ODS-Syntax für HTML-Ausgabe

Um mit ODS HTML-Output zu erzeugen, sind die folgenden Schritte nötig:

1. Das ODS Ausgabe-Ziel HTML wird geöffnet
2. Prozeduren werden submittiert
3. Das ODS Ausgabe-Zeile HTML wird geschlossen

In SAS-Syntax sieht das folgendermaßen aus:

```
ODS html path = 'c:/test' body = 'test1.html' style = statDoc;
```



```
PROC PRINT data=sashelp.class;
run;
ODS html close;
```

Die Option "path=" gibt ein Verzeichnis an, in das alle vom ODS erzeugten Dateien gespeichert werden, "body=" den Namen der zu erzeugenden HTML-Datei. „Style=" gibt einen im ODS vordefinierten Layout-Stil an.

6.1 Problem – der generierte HTML-Code

Der HTML-Code, der von SAS erzeugt wird, ist umständlich aufgebaut. Jede einzelne Zelle der Tabelle ist für sich gelayoutet, der folgende Ausschnitt definiert z.B. eine Zelle mit dem Wert 1:

```
<TD ALIGN=RIGHT bgcolor="#AAAAAA"><font face="Verdana, Helvetica, Helv"
size="2" color="#0033AA"><b> 1</b></font></TD>
```

Das bedeutet, dass die HTML-Dateien relativ groß werden. Deshalb wurde die HTML um das Konzept der **Cascading Style Sheet (CSS)** erweitert. In einem CSS werden Layout-Klassen definiert, auf die dann in einem HTML-Dokument Bezug genommen wird. Der Vorteil ist, dass das CSS nur einmal definiert (und auf den Browser übertragen) wird, während sich beliebig viele HTML-Dateien darauf beziehen können. CSS entwickeln also dann ihre Stärke, wenn viele HTML-Dateien mit dem gleichen Layout erzeugt werden sollen, für eine einzelne Datei lohnt sich das ganze nicht.

Auch das ODS unterstützt CSS, diese Option muss aber zusätzlich angegeben werden. Dann sieht das vorherige Beispiel so aus:

```
ODS html path = 'c:/test'
      body = 'test1.html' style = statDoc stylesheet = 'style.css';

PROC PRINT data=sashelp.class;
run;
ODS html close;
```

Der HTML-Code für die Zelle mit dem Wert „1“ hat sich folgendermaßen verkürzt:

```
<TD class="Header" style=" Text-align:RIGHT;"> 1</TD>
```

Leider hat auch dies einen Haken, unabhängig von den tatsächlich verwendeten CSS-Klassen, schreibt das ODS alle ihm bekannten Klassen in das CSS, d.h. auch diese Datei ist unnötig groß.

6.2 Problem – Anpassen der vordefinierten Stile

Die im ODS definierten Stile sind nicht jedermanns Sache, i.d.R. ist eine Anpassung an eigene Vorstellungen bzw. bereits existierende (Firmen-) „Corporate Designs“ notwendig. Zwar können die Stile mit Hilfe der Prozedur TEMPLATE angepasst werden, doch ist die Syntax dafür mit dem Begriff „kompliziert“ eher schmeichelhaft umschrieben. Schicken Sie einfach das folgende Programm ab, und Sie wissen, was ich meine.

```
proc template;
  source styles.default;
run;quit;
```

Die Lösung

Ich bin deshalb zu einer pragmatischen Lösung gekommen, die meiner Meinung nach einen relativ praktikablen Umgang mit ODS und HTML-Dateien ermöglicht. Sie lautet: „Lass das ODS das machen, was es gut kann, und mach den Rest selber“. Auf diese Problematik angewendet bedeutet dies:

1. die CSS-Option von ODS nutzen, denn die erzeugten CSS-Klassen sind in Ordnung
2. aber anschließend die automatisch generierte CSS-Datei mit einer eigenen Datei überschreiben, die nur die tatsächlich benötigten Klassen in der gewünschten Form enthält. Dieser Schritt kann von Hand, oder ebenfalls automatisiert aus SAS heraus erfolgen.

Ein letzter Tipp, sollen die HTML-Seiten auch einen Standard-Anfangs- und End-Block bekommen, können die ODS html Optionen „notop“ und „nobot“ genutzt werden. Mehr Informationen hierzu bietet die SAS Online Documentation.

7. Export von SAS Grafiken nach Microsoft Office 97 mit Version 8. Ein Vergleich von EMF und CGM *Heinrich Stürzl*

Einleitung

Für das Exportieren von Grafiken von SAS/GRAPH in der Version 8 nach Microsoft Powerpoint, Word, Excel 97 werden vom Technical Support die beiden Vektorgrafikformate **EMF (Enhanced Windows Metafile)** und **CGM (Computer Graphics Metafile)** empfohlen. EMF ist die Weiterentwicklung des Windows Metafile Formats (WMF) und ist ausschließlich unter Windows verfügbar. Beide Formate erlauben die Verwendung von Hardware Fonts. Bei CGM sind dies ausgewählte Schriftarten, die mit F=HWCGMnnn angesprochen werden können (siehe nächste Tabelle).

Bei **EMF**, sowie **WMF, BMP und GIF** lassen sich alle installierten True Type Fonts direkt mit ihrem Namen verwenden z. B. F="Arial Narrow". Ein Vorteil der TrueType Fonts ist, dass diese Schriftart in der Office Anwendung genauso wiedererkannt wird. Text kann dadurch nachträglich geändert werden. Bei vertikalem Text gibt es jedoch weiterhin Probleme. Er wird buchstabenweise gedreht, sobald die Datei editiert wird. Ein weiterer Vorteil der True Type Fonts gegenüber den gefüllten SAS Schriftarten (SWISS, ZAPF, ...) ist, dass die Exportdateien relativ klein und gut handhabbar bleiben. Grundsätzlich wird von SAS Institute empfohlen, EMF gegenüber CGM zu bevorzugen.

Grafiken können auch über die **Zwischenablage** aus dem Graph Fenster kopiert werden. Dies ist in der Regel mit einem Qualitätsverlust gegenüber dem Export im EMF Format (**File → Export as Image, *.emf**) verbunden.

Bei der Erzeugung von Grafikexportdateien per Programm gab es bisher (vor Version 8) Probleme, wenn mehrere Grafiken in der Datei enthalten waren. Falls zum Beispiel durch ein By- oder Run-group-processing mehrere Grafiken erzeugt werden, werden alle in die angegebene CGM Datei geschrieben. In der Office Anwendung kann jedoch jeweils nur eine Grafik importiert und angezeigt werden (ohne die "Advanced Filter Options" sogar nur die erste Grafik).

Eine Exportdatei pro Grafik automatisch erstellen

Ab Version 8 ist es möglich, anstelle einer expliziten Grafikexportdatei ein Dateiverzeichnis anzugeben, in welches pro erzeugter Grafik eine separate Exportdatei geschrieben wird. Als Dateinamen werden die Namen der GRSEG Einträge verwendet und die Dateiendung wird automatisch entsprechend dem Treiber ergänzt. Leider unterstützt SAS/GRAPH noch nicht

die langen Namen, so dass die GRSEG Einträge und damit die Namen der Exportdateien auf acht Zeichen beschränkt sind.

Die Syntax ist ansonsten unverändert.

```
FILENAME outfile "C:\temp";
GOPTIONS GSFNAME=outfile;
```

Export via CGM

Für das Erstellen von CGM-Dateien sollten die Treiber **CGMOF97L** und **CGMOF97P** mit folgender Syntax verwendet werden.

```
GOPTIONS DEV=CGMOF97L KEYMAP=WINANSI DEVMAP=WINANSI FTEXT=HWCGM001
          GSFLLEN=8092 GSFNAME=outfile;
```

Unterstützte Hardware Schriftarten in den **CGMOF97x** Treibern

Fontname in SAS	Fontname in Office 97	Fontname in the Chartype Screen
HWCGM001	Arial	Helvetica
HWCGM002	Arial Bold	HelveticaBold
HWCGM003	<i>Arial Italic</i>	<i>HelveticaItalic</i>
HWCGM004	<i>Arial Bold Italic</i>	<i>HelveticaBoldItalic</i>
HWCGM005	TimesNewRoman	TimesNewRoman
HWCGM006	TimesNewRomanBold	TimesNewRomanBold
HWCGM007	<i>TimesNewRomanItalic</i>	<i>TimesNewRomanItalic</i>
HWCGM008	<i>TimesNewRomanBoldItalic</i>	<i>TimesNewRomanBoldItalic</i>
HWCGM009	CourierNew	CourierNew
HWCGM010	CourierNewBold	CourierNewBold
HWCGM011	<i>CourierNewItalic</i>	<i>CourierNewItalic</i>
HWCGM012	<i>CourierNewBoldItalic</i>	<i>CourierNewBoldItalic</i>
HWCGM013	αβγδεφγηηθκ (Symbol)	Symbol
HWCGM014	◆(■)ⓂⓃ(■)Ⓜ◆ (wingdings)	ZapfDingbats

Von den **Linienarten** werden nur die ersten fünf SAS Linientypen unterstützt, d.h. in Office 97 als Linie erkannt. Allerdings wird die Verwendung der durchgezogenen Linie (l=1) empfohlen.

Unterstützte **Symbole** sind DOT und STAR. Für andere Zeichen wird die Schriftart Wingdings verwendet.

Export via EMF

EMF unterstützt mehr Hardware Fonts als CGM, nämlich alle jeweils verfügbaren True Type Fonts. Diese können direkt über ihren Namen angesprochen werden (Genauere Schreibweise ist wichtig). Zum Beispiel: font="Arial Narrow".

Optional können weitere Attribute nach einem "/" angehängt werden, wobei kein Unterschied zwischen Groß/Kleinschreibung besteht und Abkürzungen erlaubt sind. Zum Beispiel:

```
f="Arial Narrow/bold"
f="Arial Narrow/bo"
f="Arial Narrow/bold/italic"
f="Arial Narrow/Bo/It"
```

Für eine optimale Anzeige der Grafiken in Word und Powerpoint werden unterschiedliche Konfigurationen der HSIZE und VSIZE Grafik Optionen empfohlen, da beide Programme EMF Dateien mit unterschiedlicher Größe importieren.

```
GOPTIONS HSIZE=6    IN VSIZE=5 IN; * Export nach MS Word;
GOPTIONS HSIZE=9.5 IN VSIZE=7 IN; * Export nach MS Powerpoint;
```

Die Füllung von Flächen (PATTERN=SOLID; X1; L1; R1) werden als einzelne Objekte importiert.

Testprogramm

Mit dem nachfolgenden Testprogramm wurde die folgende Grafik mit PROC GSLIDE in Version 8.1 als EMF Datei (10 kByte) erzeugt und in Word importiert. Alle Schriften sind auf dieselbe Größe eingestellt.



```
FILENAME outfile "C:\temp\"; * Zielverzeichnis 1 Datei pro Grafik;
GOPTIONS RESET=ALL KEYMAP=WINANSI DEVMAP=WINANSI DEV=EMF
TARGET=WINPRTC ROTATE=LANDSCAPE GUNIT=CM CTEXT=BLACK
FTEXT='Arial/bold' HTEXT=0.7 CM LFACTOR=1 GSFNAME=outfile;
GOPTIONS HSIZE=6    IN VSIZE=5 IN; * Graph File Export nach MS Word;
```

```
TITLE1 F="Arial" H=0.7 CM "Arial öäü ÖÄÜ ß";
TITLE2 F="Arial/bold" H=0.7 CM "Arial/bold öäü ÖÄÜ ß";
TITLE3 F="Arial/bold/italic" H=0.7 CM "Arial/bold/italic öäü ÖÄÜ ß";
```

```
TITLE4 F="Times New Roman" H=0.7 CM "Times New Roman  öäü ÖÄÜ ß";
TITLE5 F="Times New Roman/bo" H=0.7 CM "Times New Ro./bo öäü ÖÄÜ ß";
TITLE6 F="Times New Roman/bold/italic" H=0.7 CM "Times New
Roman/bold/italic öäü ÖÄÜ ß";
TITLE7 F="Courier New" H=0.7 CM "Courier New  öäü ÖÄÜ ß";
TITLE8 F="Courier New/bold" H=0.7 CM "Courier New/bold  öäü ÖÄÜ ß";
TITLE9 F="Courier New/bold/italic" H=0.7 CM "Courier New/bold/italic
öäü ÖÄÜ ß";
TITLE10 F="Symbol" H=0.7 CM "Symbol abcdefghijklmnopqrstuvwxyz";
FOOTNOTE;
FOOTNOTE2 F="Arial Narrow" H=0.7 CM "Arial Narrow  öäü ÖÄÜ ß";
FOOTNOTE3 F="Arial Narrow/bo" H=0.7 CM "Arial Narrow/bo  öäü ÖÄÜ ß";
FOOTNOTE4 F="Arial Narrow/bold/italic" H=0.7 CM "Arial
Narrow/bold/italic öäü ÖÄÜ ß";
FOOTNOTE5 F="Wingdings" H=0.7 CM "abcdefghijklmnopqrstuvwxyz";
FOOTNOTE10 F="Arial" A=90 H=0.4 CM "&SYSVLONG &SYSDATE
%SYSFUNC(TIME( ),TIME5.)";

PROC GSLIDE; RUN; QUIT;
PROC GTESTIT; RUN;
```



Quellen

- Technical Support Document TS252X - 2nd Edition, 1. Nov. 1999, Exporting SAS/GRAPH Output to Microsoft Word, Excel and Powerpoint 97
- Technical Support Document TS-300D, "PRODUCING HARDCOPY GRAPHICS UNDER WINDOWS 95 and NT (Version 7 of SAS/GRAPH® Software)"